

ANALISI DI VOLANO ROTANTE

Si costruisca un modello FEM per l'analisi dello stato di tensione del volano rotante a velocità angolare Ω mostrato nella Fig. 1. Si approssimi il tratto curvo con un'unica spline cubica (comando BSPLINE).

Si assuma l'albero come rigido. Si verifichi l'effetto di assumere un coefficiente di attrito tra albero e volano nullo o molto grande.

Si usi il sistema mks di unità di misura ed il comando OMEGA (rad/sec) per inserire la rotazione attorno all'asse ed ottenere il calcolo delle forze centrifughe. Si faccia uso del comando PATH per rappresentare l'andamento radiale delle tensioni.

Nel seguito sono riportati lo schema del modello, l'andamento calcolato delle tensioni ed un file di ingresso.

Dati del problema:

- Modulo di Young: 210000 MPa
- Coefficiente di Poisson : 0.3
- $A=0.05$ m
- $RI=0.025$ m
- $LE=0.01$ m
- $LP=0.02$ m
- $RE=0.4$ m
- $n=3000$ giri/1' (velocità di rotazione)




Fig. 1 – Schema dell'analisi




Fig. 2 – Modello FEM




Fig. 3 – Andamento tensioni

FILE DI INGRESSO

C*****

C***

C*** VOLANO SOGGETTO A FORZE CENTRIFUGHE

C***

C*** COMANDO OMEGA

c*** GENERA UN VOLANO CON PROFILO DATO DA UNA

C*** FUNZIONE DI POTENZA $Y=B*X*(1/N)$

C*** UTILIZZA ELEMENTI ASSIALSIMMETRICI PIANI

C***

C*** UNITA' DI MISURA MKS

C***

FINISH

/CLEAR

/PREP7

C****

C**** PARAMETRI

C***

A=0.05 ! RAGGIO INIZIO RACCORDO FUNZIONE POTENZA

N=2 ! PARAMETRO "N" FUNZIONE PROFILO

RI=0.025 ! RAGGIO INTERNO

LE=0.01 ! LARGHEZZA ESTERNA

LP=0.02 ! PROFONDITA' RACCORDO PROFILO

RE=0.4 ! RAGGIO ESTERNO

$B=LP/((RE-A)**(1/N))$! PARAMETRO "B" FUNZIONE PROFILO

NY=3000 ! VELOCITA' DI ROTAZIONE (GIRI/MIN)

OMY=NY/60*(2*3.1415)!VELOCITA' DI ROTAZIONE (RAD/S)

*ASK,IVIN,VINCOLO ASSIALE 1 SOLO NODO (0) O TUTTI (1);,0

! DEFINIZIONE PROFILO PER 5 PUNTI, SPAZIATI LUNGO IL RAGGIO IN PROGRESSIONE GEOMETRICA

RAP=2 ! RAPPORTO TRA ULTIMO E PRIMO PASSO

RAG=RAP**(1/4) ! CALCOLO DELLA RAGIONE DELLA PROGRESSIONE
GEOMETRICA

$PA=(RE-A)/(1+RAG+RAG**2+RAG**3+RAG**4)$! PASSO INIZIALE

C***

C*** KEYPOINTS

C***

K,1,RI

K,2,A

PA1=PA/RAG

XX=A

*DO,IJK,1,5

PA1=PA1*RAG

XX=XX+PA1

K,IJK+2,XX,-(B*(XX-A)**(1/N))

*ENDDO

LM=LE+2*LP

LOCAL,11,0,0,-LM/2

KSYMM,Y,1,7

C***

```
C*** LINES
C***
L,1,2
BSPLIN,2,3,4,5,6,7
L,7,14
BSPLIN,14,13,12,11,10,9
L,9,8
L,8,1
```

```
C***
C*** AREA
C***
AL,ALL
```

```
C***
C*** PROPRIETA' MATERIALE
C***
MP,EX,1,2.1E11
MP,DENS,1,7850
```

```
C***
C*** ELEMENTI
C***
ET,1,42,,1
```

```
C***
C*** MESH
C***
SMRTSIZE,1
AMESH,ALL
```

```
C***
C*** VINCOLI
C***
DL,6,,UX
*IF,IVIN,EQ,1,THEN
    DL,6,,UY
*ELSE
    DK,1,UY,0
*ENDIF
```

```
C***
C*** CARICHI
C***
OMEGA,,OMY,
```

```
C***
C*** SOLUZIONE
C***
/SOLU,
SOLVE
```

```
C***
C*** POST-PROCESSING
C***
/POST1
PLDISP,1
*ASK,IFL,RETURN PER CONTINUARE,0
```

```
PLNSOL,S,X
*ASK,IFL,RETURN PER CONTINUARE,0
```

PATH,PIPP0,2,,300
PPATH,1,,RI,-LM/2
PPATH,2,,RE,-LM/2

PDEF,SRAD,S,X
PDEF,SCIR,S,Z
PDEF,SAXI,S,Y

PLPATH,SRAD,SCIR,SAXI

FILE DI INPUT PARZIALMENTE COMPILATO

```
C*****
C***
C*** ESERCITAZIONE: DIMENSIONAMENTO DI GIUNTO AD OMEGA
C***
c*** PROPRIETA' MATERIALE VARIABILI CON LA TEMPERATURA
C*** TENSIONI TERMICHE, COMBINAZIONE GRANDEZZE OTTENUTE
C*** CON IL COMANDO ETABLE
C***
FINISH
/CLEAR
C***
C*** DEFINIZIONE VARIABILI
C***
A=350 ! Distanza tra gli estremi dei tubi dritti
L=5000 ! Lunghezza complessiva tubazione
B=L/2
C=A/2
R1=100 ! rAGGIO DI CURVATURA 1

*ASK,R2,RAGGIO CURVATURA 2 :,100

D=100          ! DIAMETRO TUBO
S=2.5          ! SPESSORE TUBO
ESZ=25        ! DIMENSIONI ELEMENTI

/PREP7
C***
C*** DEFINIZIONE KEYPOINTS
C***
K,1

-----

C***
C*** DEFINIZIONE LINEE
C***
L,1,2
LARC,2,4,3,R1

-----

C***
C*** DEFINIZIONE ELEMENTI
C***
et,1,289
et,2,290,,,,,2

SECTYPE,1,PIPE,,TUBO
SECDATA,D,S

C***
C*** MATERIALE
MPTEMP,1,20,100,200,300,400,500
mp,ex,1,1.97e5,-4.79e1,4.2e-2,-3.03e-4

mp,alpx,1,0.000012

C***
C*** DEFINIZIONE VINCOLI
```

C***
DK,1,all

C***
C*** MESH
C***
LSEL,,LINE,,1
LSEL,A,LINE,,6
LATT,1,,1,,5,,1
LESIZE,ALL,ESZ

LSEL,,LINE,,2,3
LATT,1,,2,,1,,1
LESIZE,ALL,ESZ

ALLS

LMESH,ALL

C***
C*** DEFINIZIONE CARICHI
C***
tref,20
bf,all,temp,400

C***
C*** SOLUZIONE
C***

/SOLU
SOLVE

C***
C*** POST-PROCESSING
C***

/post1
/DSCALE,,1
PLDISP,1
*ASK,IFL,RETURN PER CONTINUARE,0

/ESHAPE,1
PLDISP,1
*ASK,IFL,RETURN PER CONTINUARE,0

/ESHAPE,0

ESEL,,TYPE,,2,2
ETABLE,SNI,SMISC,8
ETABLE,SNJ,SMISC,43
ETABLE,SBYI,SMISC,9
ETABLE,SBYJ,SMISC,44

ESEL,,TYPE,,1,1
ETABLE,SNI,SMISC,31
ETABLE,SNJ,SMISC,36
ETABLE,SBYI,SMISC,32
ETABLE,SBYJ,SMISC,37

ALLS
SABS,1
SADD,SI,SN,SBYI,1,1
SADD,SJ,SNJ,SBYJ,1,1

PLLS,SN,SNJ
*ASK,IFL,RETURN PER CONTINUARE,0

PLLS,SBYI,SBYJ
*ASK,IFL,RETURN PER CONTINUARE,0

PLLS,SI,SJ
*ASK,IFL,RETURN PER CONTINUARE,0

PRRFOR